*****************************************************************************;

**********************   Two sample comparison Macro   **********************;

*****************************************************************************;

/*

Notes/Documentation:

Inputs:

indata:

Name of input dataset

outdata:
Name of output dataset(contains more than the printed summary)

by:


Optional grouping variable, do analysis within each group

class:

Variable that defines the 2 classes analyzed, must be included, must have 2 levels

            Can be formatted numeric, unformatted numeric, or character

var:

Continuous variables to analyze

print:

Set to "Y" by default, specifying anything else will suppress printed summary

test:

Specify 1, 2, 3, or 4 to pick which p-value to report for test for normality of data

            1=Shapiro Wilk, 2=Kolmogorov-Smirnov, 3=Cramer-von Mises, 4=Anderson-Darling, or 




blank goes to default of Shapiro-Wilk 

lines:      Specify 1 or 3 to choose the number of lines(observations) on the output dataset per variable

            1 line per gives more variables, ie: mean_0(difference) mean_1(group 1) mean_2(group 2)

            3 lines per gives more lines, ie: 1 variable named mean with 3 lines: group 1, group 2, difference

**WARNING:** With a sample size greater than 2000, the Shapiro-Wilk test is not produced, 

this will lead to errors in the output

Notes:

-To run this macro, must also have the files for %unisum, %normaltest, & %wilcoxon. 

 Must also then update the paths in the %include statements that call those files.

-Output variable suffixes for 1 line option:

 0 denotes statistics describing the difference of the 2 levels

 1 denotes the first level when sorted ascending

 2 denotes the second level when sorted ascending

Modification/Update notes:

9/5/2014: General 'cleaning-up' update of the macro: fixed 3-line by variable printing issue, 

 1-line output label printing issue, changed drop/keep statements to avoid uninitialized variables

5/11/2010: Update of the macro with alternate output dataset style: 1 row for each sample 

 and 1 row for the difference, so a printed table can be less wide

1/30/2007: modification of %twosample to %twosample_medq:

runs %unisum2 to include the median, q1 and q3 in the final output dataset

Have not tested if it works with by variables though...?, still works with 1 2 level class variable

1/31/2007: modification to use the smaller of 2 class level normal test p-values

also have not tested that with by variables

*/

/***

example calls:

-Include automatic pring with Shapiro-Wilk normal test and 3-lined output:

%twosample2010(indata=deid, outdata=test_out4, by=sex, class=group_qc, var=age max_ftnd min_ftnd 

n_clin_foll num_fu quannic, print=Y, test=1, lines=3);

-This gives the same result, but the automatic defaults for print, test & lines do not need to be specified:

%twosample2010(indata=deid, outdata=test_out4, by=sex, class=group_qc, var=age max_ftnd min_ftnd 

n_clin_foll num_fu quannic);

-Prints automatically, but gives 1-lined output with no by-variable and uses Kolmogorov-Smirnov for normality testing:

%twosample2010(indata=deid, outdata=test_out4, class=group_qc, var=age, test=2, lines=1);

-Suppresses automatic printed output, separate proc print statements to print the named output dataset:

%twosample2010(indata=deid, outdata=test_out4, by=sex, class=group_qc, var=age, print=N, test=1, lines=3);

proc print data=test_out4 noobs;

proc print data=test_out4 noobs l;

run;

***/

*-------------------------------------------------------------------------------*;

*---------------------------TWOSAMPLE macro -- START----------------------------*;

%macro twosample(indata=, outdata=, by=, class=, var=, print=Y, test=1, lines=3);

*-------------------------------------------------------------------------------*;

*including macros used in producing output for the final, combined output dataset;

%include 'C:\Users\thostensonjeffd\Documents\work\sas_macros\macro_wilcoxon.sas';

%include 'C:\Users\thostensonjeffd\Documents\work\sas_macros\macro_normaltest.sas';

%include 'C:\Users\thostensonjeffd\Documents\work\sas_macros\macro_unisum.sas';

run;

*-------------------------------------------------------------------------------*;

*-------------------------------------------------------------------------------*;

*Defining format for p-value outputs;

proc format;

value pvalue 0 - <0.001='<0.001';

run;

*-------------------------------------------------------------------------------*;

*-------------------------------------------------------------------------------*;

*Subsetting to subjects with a value for the class variable*;

data _2s_tempdata1(drop= _temp);

set &indata;

_temp=compress(&class,' ');

if &class ne ' ';

*-------------------------------------------------------------------------------*;

run; ods listing close; run; /* Close output to listing */

*-------------------------------------------------------------------------------*;

*Run proc contents to get output dataset with analysis variable labels & formats*;

proc contents data=_2s_tempdata1; 

ods output Variables=_contents;

data _contents2(drop= Len Member Num Pos);*(keep= variable label format type); 

set _contents;

if variable=compress("&class",' ') then do; 

 call symput('classformat',format); 

 call symput('classtype',type); 

end;

run;

*-------------------------------------------------------------------------------*;

*-------------------------------------------------------------------------------*;

*If Class variable is numeric, convert to character;

*-Case 1: variable 1 is a formatted numeric variable-*;

%if &classtype=Num & &classformat^= %then %do;

*rename input numeric variable to make way for character-coerced version;

data _2s_tempdata1; set _2s_tempdata1; rename &class=_temp1; run;

*Create character-coerced version of numeric input variable;

data _2s_tempdata1;

set _2s_tempdata1;

&class=put(_temp1,&classformat);

run;

%end;

*-Case 1: variable 1 is a formatted numeric variable-*;

*-Case 2: variable 1 is an unformatted numeric variable-*;

%if &classtype=Num & &classformat=  %then %do;

data _2s_tempdata1; set _2s_tempdata1; rename &class=_temp1; run;

*Create character-coerced version of numeric input variable;

data _2s_tempdata1;

set _2s_tempdata1;

&class=compress(_temp1,' ');

run;

%end;

*-Case 2: variable 1 is an unformatted numeric variable-*;

*-------------------------------------------------------------------------------*;

*-------------------------------------------------------------------------------*;

*Running proc ttest to get summary statistics, t-test, & equal variance test output;

%if &by^=  %then %do;

proc sort data=_2s_tempdata1;

by &by;

%end;

proc ttest data=_2s_tempdata1;

by &by;

class &class;

var &var;

ods output Statistics=_stats TTests=_ttest equality=_eqvar;

data _stats;

set _stats;

order=mod(_n_,3);

run;

*-------------------------------------------------------------------------------*;

*-------------------------------------------------------------------------------*;

*Running proc freq to get class level labels;

proc freq data=_2s_tempdata1;

tables &class;

ods output OneWayFreqs=_1wf;

proc transpose data=_1wf out=_t_1wf prefix=level_;

var &class;

data _t_1wf;

set _t_1wf;

rename _name_=class;

dummy=1;

run;

*-------------------------------------------------------------------------------*;

*-------------------------------------------------------------------------------*;

*Running %unisum to output median, q1 and q3 for summary output with the rank sum test;

%unisum(indata=_2s_tempdata1, outdata=_unisum, var_list=&var, by=&by &class, output=median=median q1=q1 q3=q3, print=N);

data _unisum;

set _unisum;

order=mod(_n_+1,2)+1;*tobacco=level 1 mj=level 2... bys & classes?... set up for 2 levels of class;

*Double transposing stats to make 1 row for each byvalue/var combination;

proc sort data=_unisum; by &by variable &class order;

proc transpose data=_unisum out=_t_unisum;

by &by variable &class order;

var q1 median q3;

run;

data _t_unisum(drop= _name_ _label_ );

set _t_unisum;

length _id $ 50;

_id=compress(_name_,' ')||'_'||compress(order,' ');

run;

proc sort data=_t_unisum; by &by variable;

proc transpose data=_t_unisum out=_tt_unisum;

by &by variable;

id _id;

var col1;

data _tt_unisum(keep= &by variable median_1 median_2 q1_1 q1_2 q3_1 q3_2);*(drop= _name_ _label_);

set _tt_unisum;

run;

*-------------------------------------------------------------------------------*;

*-------------------------------------------------------------------------------*;

*Double transposing stats to make 1 row for each byvalue/var combination;

proc sort data=_stats; by &by variable class order;

proc transpose data=_stats out=_t_stats;

by &by variable class order;

var LowerCLMean LowerCLStdDev Maximum Mean Minimum N StdDev StdErr UMPULowerCLStdDev 

UMPUUpperCLStdDev UpperCLMean UpperCLStdDev;

data _t_stats(drop= _name_ _label_ class);

set _t_stats;

length _id $ 50;

_id=compress(_name_,' ')||'_'||compress(order,' ');

if col1<=. then delete;

proc sort data=_t_stats; by &by variable;

proc transpose data=_t_stats out=_tt_stats;

by &by variable;

id _id;

var col1;

data _tt_stats(drop= _name_);

set _tt_stats;

run;

*-------------------------------------------------------------------------------*;

*-------------------------------------------------------------------------------*;

*Double transposing ttest to make 1 row for each byvalue/var combination;

proc sort data=_ttest; by &by variable variances;

proc transpose data=_ttest out=_t_ttest;*ignore method variable;

by &by variable variances;

var tValue DF Probt;

data _t_ttest(drop= _name_ _label_ );

set _t_ttest;

length _id $ 50;

_id=compress(_name_,' ')||'_'||compress(variances,' ');

proc sort data=_t_ttest; by &by variable;

proc transpose data=_t_ttest out=_tt_ttest;

by &by variable;

id _id;

var col1;

data _tt_ttest(drop= _name_);

set _tt_ttest;

run;

*-------------------------------------------------------------------------------*;

*-------------------------------------------------------------------------------*;

*Equal variance test output;

data _eqvar(drop= method);

set _eqvar;

rename NumDF=NumDF_eqvar DenDF=DenDF_eqvar FValue=FValue_eqvar ProbF=ProbF_eqvar;

run;

*-------------------------------------------------------------------------------*;

run; ods listing; run; /* re-open output to listing (have to re-open here, other macros close & re-open too)*/

*-------------------------------------------------------------------------------*;

*Call %wilcoxon and %normaltest macros to do rank sum test & test for normality;

%wilcoxon(indata=_2s_tempdata1, outdata=_wilcox, var=&var, class=&class, by=&by, print=N);

run;

%normaltest(indata=_2s_tempdata1, outdata=_normaltest, var=&var, by=&by &class, test=&test, print=N); run;

/*

Retain the smallest of the 2 p-values from normal-testing the 2 populations.

If the smallest fails/passes, then normality for the 2-sample t-test fails/passes respectively.

*/

proc sort data=_normaltest; by &by varname pvalue_normal;

data _normaltest1(keep= &by &class VarName p_normal psign_normal pvalue_normal);*(drop= group01);

set _normaltest;

by &by varname pvalue_normal;

if first.varname;

run;

data _normaltest1;

set _normaltest1;

rename varname=variable;

run;

*-------------------------------------------------------------------------------*;

*-------------------------------------------------------------------------------*;

*Combine output datasets into 1, label variables;

*Merge the various output datasets: by variable & &by;

proc sort data=_tt_stats; by variable &by;

proc sort data=_tt_unisum; by variable &by;

proc sort data=_tt_ttest; by variable &by;

proc sort data=_eqvar; by variable &by;

proc sort data=_wilcox; by variable &by;

proc sort data=_normaltest1; by variable &by;

data _out_merged;*(drop= _name_);

length variable $ 32;

merge _tt_stats _tt_unisum _tt_ttest _eqvar _wilcox _normaltest1;

by variable &by;

dummy=1;

run;

*Merge on contents output: variable names with variable labels: by variable;

proc sort data=_out_merged; by variable;

proc sort data=_contents2; by variable;

data _out_merged;

merge _contents2 _out_merged(in=in1);

by variable;

if in1;

run;

*Labels, & merge on class variable name, label, and level values: by dummy;

proc sort data=_out_merged; by dummy;

proc sort data=_t_1wf; by dummy;

data _outdata1(drop= dummy);

merge _out_merged _t_1wf;

by dummy;

label Variable='Analysis variable' class='Class variable' level_1='Level 1' level_2='Level 2'

Mean_0='Mean, difference' Mean_1='Mean, level 1' Mean_2='Mean, level 2' N_1='n, level 1' N_2='n, level 2' 

StdDev_0='Std Dev, difference' StdDev_1='Std Dev, level 1' StdDev_2='Std Dev, level 2' 

StdErr_0='Std Error, difference' StdErr_1='Std Error, level 1' StdErr_2='Std Error, level 2'

median_1='Median, level 1' median_2='Median, level 2'

q1_1='25th percentile, level 1' q1_2='25th percentile, level 2'

q3_1='75th percentile, level 1' q3_2='75th percentile, level 2' 

Maximum_1='Maximum, level 1' Maximum_2='Maximum, level 2' 

Minimum_1='Minimum, level 1' Minimum_2='Minimum, level 2' 

LowerCLMean_0='Lower Limit of Mean, difference' 

LowerCLMean_1='Lower Limit of Mean, level 1' 

LowerCLMean_2='Lower Limit of Mean, level 2' 

LowerCLStdDev_0='Lower Limit of Std Dev, difference' 

LowerCLStdDev_1='Lower Limit of Std Dev, level 1' 

LowerCLStdDev_2='Lower Limit of Std Dev, level 2' 

UMPULowerCLStdDev_0='UMPU Lower Limit of Std Dev, difference' 

UMPULowerCLStdDev_1='UMPU Lower Limit of Std Dev, level 1' 

UMPULowerCLStdDev_2='UMPU Lower Limit of Std Dev, level 2' 

UMPUUpperCLStdDev_0='UMPU Upper Limit of Std Dev, difference' 

UMPUUpperCLStdDev_1='UMPU Upper Limit of Std Dev, level 1' 

UMPUUpperCLStdDev_2='UMPU Upper Limit of Std Dev, level 2' 

UpperCLMean_0='Upper Limit of Mean, difference' 

UpperCLMean_1='Upper Limit of Mean, level 1' 

UpperCLMean_2='Upper Limit of Mean, level 2' 

UpperCLStdDev_0='Upper Limit of Std Dev, difference' 

UpperCLStdDev_1='Upper Limit of Std Dev, level 1' 

UpperCLStdDev_2='Upper Limit of Std Dev, level 2' 

DF_Equal='DF, equal var. t-test' DF_Unequal='DF, unequal var. t-test' 

Probt_Equal='p-value, equal var. t-test' Probt_Unequal='p-value, unequal var. t-test' 

tValue_Equal='t-value, equal var. t-test' tValue_Unequal='t-value, unequal var. t-test'

DenDF_eqvar='Denominator Degrees Freedom' FValue_eqvar='F Value'

NumDF_eqvar='Numerator Degrees Freedom' ProbF_eqvar='Pr > |F|';

run;

*-------------------------------------------------------------------------------*;

*-------------------------------------------------------------------------------*;

*Test/P-value picking code;

/**

adds four variables:  pvalout pvaltest pvalnonp pvalsig

pvalout: numeric, appropriate p-value to report based on normality and

    equal variance assumption tests

pvaltest: character, which pvalue was chosen for pvalout,

    Rank Sum test or t-test, unequal variances or t-test, equal variances

pvalnonp: character, pvalout with asterisk denoting observations

    using the non-parametric Rank Sum test

pvalsig: character, pvalout with asterisk denoting

    significant(p<=.05) p-values

**/

*Pick main test/p-value to output in short summary;

data _outdata1;

set _outdata1;

    *-----pvalout and pvaltest code-----*;

    length pvaltest $ 25;

    *Loop for rejecting normality and using the rank sum test;

    if 0<=pvalue_normal<=0.05 & psign_normal in('<',' ')

        then do;

        pvalout=p2_wil;*put(p2_wil,8.4);*normal approximation, 2-sided;

        pvaltest='Rank Sum test';

        end;

    *Loop for accepting normality, then testing for equal variances, use t-test;

    else if pvalue_normal>0.05 or (pvalue_normal=0.05 & psign_normal='>') then do;

        if probf_eqvar<=.05 then do;

            pvalout=probt_unequal;*put(probt_unequal,8.4);

            pvaltest='t-test, unequal variances';

            end;

        else do;

            pvalout=probt_equal;*put(probt_equal,8.4);

            pvaltest='t-test, equal variances';

            end;

        end;

    else pvalout=.e;*check values of psign_normal, pvalue_normal, p_normal test;

    *-----pvalout and pvaltest code-----*;

    *-----pvalnonp code-----*;

    if pvalout ne .e then do;

        if pvaltest='Rank Sum test' then

           pvalnonp=compress((put(round(pvalout,0.0001),pvalue.)),' ')||'*';

         else pvalnonp=put(round(pvalout,0.0001),pvalue.);

      end;

    *-----pvalnonp code-----*;

    /*-----pvalsig code-----*;


 *alternate version to give 3*s to under .05 and 1* to .05 to .10;

    if .<pvalout<=.05 then

        pvalsig=compress((put(round(pvalout,0.0001),pvalue.)),' ')||'***';

     else if .05<pvalout<=.1 then

        pvalsig=compress((put(round(pvalout,0.0001),pvalue.)),' ')||'*';

     else if .1<pvalout then

        pvalsig=compress((put(round(pvalout,0.0001),pvalue.)),' ');

      else if pvalout=. then pvalsig=' ';

    *-----pvalsig code-----*/;

    *-----pvalsig code-----*;

    if pvalout ne .e then do;

     if .<pvalout<=.05 then

        pvalsig=compress((put(round(pvalout,0.0001),pvalue.)),' ')||'*';

      else if .05<pvalout then

        pvalsig=compress((put(round(pvalout,0.0001),pvalue.)),' ');

       else if pvalout=. then pvalsig=' ';


end;

    *-----pvalsig code-----*;

    label pvalout='p-value' pvaltest='Test' pvalnonp='p-value (non-param.)'

        pvalsig='p-value';


run;

*-------------------------------------------------------------------------------*;

*-------------------------------------------------------------------------------*;

*Create alternate output dataset: 3 rows for each variable: sample 1, sample 2, difference;

*-Sample 1 dataset-*;

data _level1(keep= LowerCLMean_1 LowerCLStdDev_1 Maximum_1 Mean_1 Minimum_1 N_1 StdDev_1 StdErr_1 

 UMPULowerCLStdDev_1 UMPUUpperCLStdDev_1 UpperCLMean_1 UpperCLStdDev_1 median_1 q1_1 q3_1 level_1 ExpectedSum_0

 MeanScore_0 StdDevOfSum_0 SumOfScores_0       Variable class 

 DF_Equal DF_Unequal DenDF_eqvar FValue_eqvar NumDF_eqvar P2_WIL PR_WIL PT2_WIL PTR_WIL ProbF_eqvar

 Probt_Equal Probt_Unequal Z_WIL _WIL_ p_normal psign_normal pvalnonp pvalout pvalsig pvaltest

pvalue_normal tValue_Equal tValue_Unequal &by);

*(keep= LowerCLMean_1 LowerCLStdDev_1 Maximum_1 Mean_1 Minimum_1 N_1 StdDev_1 StdErr_1 

 UMPULowerCLStdDev_1 UMPUUpperCLStdDev_1 UpperCLMean_1 UpperCLStdDev_1 median_1 q1_1 q3_1 level_1 ExpectedSum_0

 MeanScore_0 StdDevOfSum_0 SumOfScores_0       Variable _LABEL_ class 

 DF_Equal DF_Unequal DenDF_eqvar FValue_eqvar NumDF_eqvar P2_WIL PR_WIL PT2_WIL PTR_WIL ProbF_eqvar

 Probt_Equal Probt_Unequal Z_WIL _WIL_ p_normal psign_normal pvalnonp pvalout pvalsig pvaltest

pvalue_normal tValue_Equal tValue_Unequal &by);

set _outdata1;

run;

data _level1;

set _level1;

sample=1;

rename LowerCLMean_1=LowerCLMean LowerCLStdDev_1=LowerCLStdDev Maximum_1=Maximum Mean_1=Mean Minimum_1=Minimum 

 N_1=n StdDev_1=StdDev StdErr_1=StdErr UMPULowerCLStdDev_1=UMPULowerCLStdDev 

 UMPUUpperCLStdDev_1=UMPUUpperCLStdDev UpperCLMean_1=UpperCLMean UpperCLStdDev_1=UpperCLStdDev median_1=median 

 q1_1=q1 q3_1=q3 level_1=level ExpectedSum_0=ExpectedSum MeanScore_0=MeanScore StdDevOfSum_0=StdDevOfSum 

 SumOfScores_0=SumOfScores;

run;

*-Sample 1 dataset-*;

*-Sample 2 dataset-*;

data _level2(keep= LowerCLMean_2 LowerCLStdDev_2 Maximum_2 Mean_2 Minimum_2 N_2 StdDev_2 StdErr_2 

 UMPULowerCLStdDev_2 UMPUUpperCLStdDev_2 UpperCLMean_2 UpperCLStdDev_2 median_2 q1_2 q3_2 level_2 ExpectedSum_1

 MeanScore_1 StdDevOfSum_1 SumOfScores_1       Variable class &by);

*(keep= LowerCLMean_2 LowerCLStdDev_2 Maximum_2 Mean_2 Minimum_2 N_2 StdDev_2 StdErr_2 

 UMPULowerCLStdDev_2 UMPUUpperCLStdDev_2 UpperCLMean_2 UpperCLStdDev_2 median_2 q1_2 q3_2 level_2 ExpectedSum_1

 MeanScore_1 StdDevOfSum_1 SumOfScores_1       Variable _LABEL_ class &by);

set _outdata1;

data _level2;

set _level2;

sample=2;

rename LowerCLMean_2=LowerCLMean LowerCLStdDev_2=LowerCLStdDev Maximum_2=Maximum Mean_2=Mean Minimum_2=Minimum 

 N_2=n StdDev_2=StdDev StdErr_2=StdErr UMPULowerCLStdDev_2=UMPULowerCLStdDev 

 UMPUUpperCLStdDev_2=UMPUUpperCLStdDev UpperCLMean_2=UpperCLMean UpperCLStdDev_2=UpperCLStdDev median_2=median 

 q1_2=q1 q3_2=q3 level_2=level ExpectedSum_1=ExpectedSum MeanScore_1=MeanScore StdDevOfSum_1=StdDevOfSum 

 SumOfScores_1=SumOfScores;

run;

*-Sample 2 dataset-*;

*-Difference dataset-*;

data _level3(keep= LowerCLMean_0 LowerCLStdDev_0 Mean_0 StdDev_0 StdErr_0 UMPULowerCLStdDev_0 

 UMPUUpperCLStdDev_0 UpperCLMean_0 UpperCLStdDev_0             Variable class &by);

*(keep= LowerCLMean_0 LowerCLStdDev_0 Mean_0 StdDev_0 StdErr_0 UMPULowerCLStdDev_0 

 UMPUUpperCLStdDev_0 UpperCLMean_0 UpperCLStdDev_0             Variable _LABEL_ class &by);

set _outdata1;

data _level3;

set _level3;

sample=3;

rename LowerCLMean_0=LowerCLMean LowerCLStdDev_0=LowerCLStdDev Mean_0=Mean StdDev_0=StdDev StdErr_0=StdErr 

 UMPULowerCLStdDev_0=UMPULowerCLStdDev UMPUUpperCLStdDev_0=UMPUUpperCLStdDev UpperCLMean_0=UpperCLMean 

 UpperCLStdDev_0=UpperCLStdDev;

run;

*-Difference dataset-*;

*-Re-set together-*;

data _outdata3;

set _level1 _level2 _level3;

length text $ 15;

if sample=3 then text='Difference';

 else text=level;*text=compress(put(level,&classformat),' ');

format mean 8.2 stddev 8.2;

label ExpectedSum='Expected Under H0 (Wilcox)' LowerCLMean='Lower Limit of Mean'

 LowerCLStdDev='Lower Limit of Std Dev' MeanScore='Mean Score (Wilcox)' StdDevOfSum='Std Dev Under H0 (Wilcox)'

 StdErr='Std Error' SumOfScores='Sum of Scores (Wilcox)' UMPULowerCLStdDev='UMPU Lower Limit of Std Dev'

 UMPUUpperCLStdDev='UMPU Upper Limit of Std Dev' UpperCLMean='Upper Limit of Mean'

 UpperCLStdDev='Upper Limit of Std Dev' q1='25th percentile' q3='75th percentile' n='n' mean='Mean' 

 stddev='Std. Dev.' median='Median' minimum='Minimum' maximum='Maximum' pvaltest='Test' pvalsig='p-value' 

 level='Class Var. Level' Variable='Analysis Variable' class='Class Variable' sample='Sample' text="&class";

 *_LABEL_='By-group Variable';

run;

*-Re-set together-*;

*-Merge labels from contents onto output dataset-*;

proc sort data=_contents2; by variable;

proc sort data=_outdata3; by variable;

data _outdata3;

merge _outdata3(in=in1) _contents2;

by variable;

if in1;

run;

*-Merge labels from contents onto output dataset-*;

*-------------------------------------------------------------------------------*;

*-------------------------------------------------------------------------------*;

*Create output dataset & print output based on print and lines options specified in macro call;

**--------**;

%if &lines=1  %then %do;

data &outdata;

set _outdata1;

run;

*----*;

*Optional printed output;

%if &print=Y %then %do;

proc sort data=&outdata; by variable &by;

proc print data=&outdata noobs l;

by variable &by;

id variable &by;

var class level_1 level_2 mean_0 n_1 mean_1 n_2 mean_2 pvaltest pvalsig;

*var class level_1 level_2 mean_0 lowerclmean_0 upperclmean_0 n_1 mean_1 n_2 mean_2 


p_normal p2_wil probf_eqvar probt_equal probt_unequal pvaltest pvalsig;

title; run; title;

%end;

*----*;

%end;

**--------**;

**--------**;

%if &lines=3  %then %do;

data &outdata;

set _outdata3;

run;

*----*;

*Optional printed output;

%if &print=Y %then %do;

proc sort data=&outdata; by variable &by sample;

proc print data=&outdata l; 

by variable &by;

id variable &by;

var text n mean stddev median minimum maximum pvaltest pvalsig; 

 *p_normal p2_wil probf_eqvar probt_equal probt_unequal;

title; run; title;

%end;

*----*;

%end;

**--------**;

*-------------------------------------------------------------------------------*;

*-------------------------------------------------------------------------------*;

*Delete temporary working datasets in macro;

proc datasets library=work nolist nowarn;

delete _out_merged _stats _ttest _eqvar _t_stats _t_unisum _t_ttest _tt_stats _tt_unisum _tt_ttest _normaltest 

 _wilcox _outdata1 _outdata3;

run;

quit;

*-------------------------------------------------------------------------------*;

%mend;

run;

*----------------------------TWOSAMPLE macro -- END-----------------------------*;

*-------------------------------------------------------------------------------*;
